
Subject: Re: Image and Maps with IDL8

Posted by chris_torrence@NOSPAM on Wed, 22 Sep 2010 20:46:10 GMT

[View Forum Message](#) <> [Reply to Message](#)

Hi David,

How about something like the following code? Obviously I just made up the "data" array - you would want to substitute in your own warped data.

I think the key is that I don't set the x/yrange until the very end. It looks like we might have a bug where it keeps resetting the ranges when you overplot things like images, continents, etc.

Cheers,
Chris
ITTVIS

```
;-----  
xorigin = -2400000.0D  
yorigin = -1440000.0D  
xend = xorigin + (9600L * 500)  
yend = yorigin + (5760L * 500)  
data = SMOOTH(RANDOMU(s,960,576),50, /EDGE_TRUNCATE)  
  
m = MAP('Lambert Azimuthal', SPHERE_RADIUS=6371228L, $  
  CENTER_LATITUDE=90.0, CENTER_LONGITUDE=0.0, $  
  XRange=[xorigin,xend], YRange=[yorigin,yend], $  
  MARGIN=[0,0,0,0], DIMENSIONS=[960,576])  
  
; Tweak the grid properties  
grid = m.MapGrid  
grid.GRID_LATITUDE = 5  
grid.GRID_LONGITUDE = 30  
grid.LABEL_POSITION = 0.86  
grid.TRANSPARENCY = 50  
  
im = IMAGE(data, $  
  IMAGE_DIMENSIONS=[xend-xorigin,yend-yorigin], $  
  IMAGE_LOCATION=[xorigin,yorigin], $  
  GRID_UNITS='meters', /OVERPLOT, RGB_TABLE=33)  
  
m1 = MAPCONTINENTS(/FILL_BACKGROUND, TRANSPARENCY=50)  
  
; Set the range at the end, once the continents are drawn.  
m.xrange = [xorigin,xend]  
m.yrange = [yorigin,yend]
```
